Oszillator-Wellenfunktionen und Kern des $n \times n$ GUE

$$\phi[n\_, x\_] := \frac{\text{HermiteH}[n, x]}{\sqrt{\sqrt{\pi}\, n!\, 2^n}}\, \text{Exp}\left[-x^2/2\right]$$

```
Table[Integrate[ϕ[j, x] ϕ[k, x], {x, -∞, ∞}],
  {j, 0, 5}, {k, 0, 5}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

$$K[\phi\_, n\_, x\_, y\_] := \sqrt{\frac{n}{2}}\, \frac{\phi[n, x]\, \phi[n-1, y] - \phi[n-1, x]\, \phi[n, y]}{x - y};$$

$$K[\phi\_, n\_, x\_, x\_] := \sqrt{\frac{n}{2}}\, \left(\phi[n-1, x]\, \phi^{(0,1)}[n, x] - \phi[n, x]\, \phi^{(0,1)}[n-1, x]\right);$$

$$K1[\phi\_, n\_, x\_, y\_] := \frac{1}{2}\, \frac{\phi[n, x]\, \phi^{(0,1)}[n, y] - \phi^{(0,1)}[n, x]\, \phi[n, y]}{x - y} - \frac{1}{2}\, \phi[n, x]\, \phi[n, y];$$

$$K1[\phi\_, n\_, x\_, x\_] := \frac{1}{2}\, \left(\phi^{(0,1)}[n, x]^2 - \phi[n, x]^2 - \phi[n, x]\, \phi^{(0,2)}[n, x]\right);$$

```
{FullSimplify[K[ϕ, n, x, y] == K1[ϕ, n, x, y], n ≥ 1],
 FullSimplify[K[ϕ, n, x, x] == K1[ϕ, n, x, x], n ≥ 1]}
```

```
{True, True}
```

```
Plot[ϕ[50, x], {x, -1.5 Sqrt[100], 1.5 Sqrt[100]}]
```



- **Plancherel-Rotach-Asymptotik**

```
With[{n = 20}, Plot[{ϕ[n, Sqrt[2 n + 1] Cos[t]],
    (2^(1/4) n^(-1/4))/Sqrt[π] Sin[(n/2 + 1/4) (Sin[2 t] - 2 t) + (3 π)/4]/Sqrt[Sin[t]]},
  {t, ArcCos[1 - (2^(-1/2) n^(-1/6) 2)/Sqrt[2 n + 1]], π - ArcCos[1 - (2^(-1/2) n^(-1/6) 2)/Sqrt[2 n + 1]]}]]
```



$$\texttt{With}\left[\{n=20\},\ \texttt{Plot}\left[\left\{\phi\left[n,\ \sqrt{2n+1}+2^{-1/2}\,n^{-1/6}\,t\right],\ 2^{1/4}\,n^{-1/12}\,\texttt{AiryAi}[t]\right\},\ \{t,-5,5\}\right]\right]$$



$$\texttt{With}\left[\{n=20\},\ \texttt{LogPlot}\left[\left\{\phi\left[n,\ \sqrt{2n+1}\ \texttt{Cosh}[t]\right],\ \frac{2^{-3/4}\,n^{-1/4}}{\sqrt{\pi}}\ \frac{\texttt{Exp}\left[\left(\frac{n}{2}+\frac{1}{4}\right)(2t-\texttt{Sinh}[2t])\right]}{\sqrt{\texttt{Sinh}[t]}}\right\},\ \left\{t,\ \texttt{ArcCosh}\left[1+\frac{2^{-1/2}\,n^{-1/6}\,2}{\sqrt{2n+1}}\right],\ 1\right\},\ \texttt{PlotRange}\to\texttt{Full}\right]\right]$$



Wigner'scher Halbkreissatz

```
ψ[n_, x_] =
  FullSimplify[(2^(1/4) n^(-1/4))/Sqrt[π] Sin[(n/2 + 1/4) (Sin[2 t] - 2 t) + 3π/4]/Sqrt[Sin[t]] /.
    Solve[Sqrt[2 n + 1] Cos[t] == x, t][[2]], {Abs[x] < Sqrt[2 n + 1], n ≥ 1}];
s1 = Series[1/Sqrt[n] K[ψ, n, x Sqrt[n], x Sqrt[n]], {n, ∞, 0}];
Limit[FullSimplify[Normal[s1], {n ≥ 2, 0 < x < 1}], n → ∞]
```

$$\frac{\sqrt{2-x^2}}{\pi}$$

```
ψ[n_, x_] = FullSimplify[(2^(-3/4) n^(-1/4))/Sqrt[π] Exp[(n/2 + 1/4) (2 t - Sinh[2 t])]/Sqrt[Sinh[t]] /.
    Solve[Sqrt[2 n + 1] Cosh[t] == x, t][[2]], {Abs[x] > Sqrt[2 n + 1], n ≥ 1}];
s1 = Series[1/Sqrt[n] K[ψ, n, x Sqrt[n], x Sqrt[n]], {n, ∞, 1}];
Limit[
 FullSimplify[(s3 = ((s2 = Normal[s1]) /. ⅇ^- → 1)), {x > Sqrt[2]}], n → ∞]
```

0

```
With[{n = 100}, Plot[{1/Sqrt[n] K[ϕ, n, Sqrt[n] x, Sqrt[n] x], 1/π Sqrt[Max[2 - x^2, 0]]},
  {x, -1.1 Sqrt[2], 1.1 Sqrt[2]}, PlotRange → Full]]
```



Bulk-Sacling Limit

```
ψ[n_, x_] =
  FullSimplify[ (2^(1/4) n^(-1/4))/Sqrt[π]  Sin[(n/2 + 1/4) (Sin[2 t] - 2 t) + 3π/4]/Sqrt[Sin[t]] /.
    Solve[Sqrt[2 n + 1] Cos[t] == x, t][[2]], {Abs[x] < Sqrt[2 n + 1], n ≥ 1}];
Limit[π/Sqrt[2 n] K[ψ, n, π x/Sqrt[2 n], π y/Sqrt[2 n]], n → ∞]
```

$$\frac{\sin(\pi (x - y))}{\pi x - \pi y}$$

Edge-Sacling Limit

```
ψ[n_, x_] = 2^(1/4) n^(-1/12) AiryAi[(x - Sqrt[2 n])/(2^(-1/2) n^(-1/6))];
```

```
Limit[
  2^(-1/2) n^(-1/6) K1[ψ, n, Sqrt[2 n] + 2^(-1/2) n^(-1/6) x, Sqrt[2 n] + 2^(-1/2) n^(-1/6) y], n → ∞]
```

$$\frac{\text{Ai}(x)\,\text{Ai}'(y) - \text{Ai}'(x)\,\text{Ai}(y)}{x - y}$$